

```

*****
*                               MODIFIED RANKIN SCORE - STACKED BAR CHARTS
*****
gen mrs1=mrs

#delimit ;
graph hbar (count) mrs1 , over(mrs) over(trt)
    stack
    percent
    asy
    legend(off)
    ytitle(Percentage , m(t+2) )
    title(Modified Rankin Score at last rating , size(medium) )
    ylabel(0"0%"20"20%"40"40%"60"60%"80"80%"100"100%" , labsize(small) )
    xlabel(name , pos(center) )
    intensity(*0.5)
    bar(1, bcol(green*0.9) )
    bar(2, bcol(green*0.7) )
    bar(3, bcol(green*0.4) )
    bar(4, bcol(red*0.4) )
    bar(5, bcol(red*0.6) )

    saving(mrs1 , replace)
;

contract trt mrs

#delimit ;
graph hbar _freq , over(mrs) over(trt)
    asy
    percent
    stack
    legend(off)
    intensity(*0.5)
    xlabel(name , pos(cent) )
    ytitle( "Percentage" , m(t+2) )
    saving(mrs2 , replace)
;
#delimit cr
pause

```

```
*****
*                LIPID FIGURE FOR JOINT BRITISH GUIDELINES
*****
```

```
#delimit ;
twoway (scatter risk ldl if prev==1 & arm==1 , ms(Oh) mc(red) )
       (scatter risk ldl if prev==2 & arm==1 , ms(Th) mc(dknavy) )
       (scatter risk ldl if prev==3 & arm==1 , ms(Sh) mc(green) )
       (scatter risk ldl if prev==1 & arm==2 , ms(O) mc(red) )
       (scatter risk ldl if prev==2 & arm==2 , ms(T) mc(dknavy) )
       (scatter risk ldl if prev==3 & arm==2 , ms(S) mc(green) )
       (scatter risk ldl if prev==1 & ldl~=1 & trial~=11 & trial~=16 & trial~=18
        , ms(i) c(L) clc(red) )
       (scatter risk ldl if prev==1 & ldl~=1 & (trial==11 | trial==16)
        , ms(i) c(L) clc(red) )
       (scatter risk ldl if prev==2 & ldl~=1 , ms(i) c(L) clc(dknavy) )
       (scatter risk ldl if prev==3 & ldl~=1 , ms(i) c(L) clc(green) )
       (scatter risk ldl if arm==2 & trial~=13 & trial~=6 & trial~=17 & trial~=18
        , ms(i) mlabsize(vsmall) mlabcol(gs2)
        mlab(trial) mlabpos(3) )
       (scatter risk ldl if arm==2 & (trial==13 | trial==6)
        , ms(i) mlabsize(vsmall) mlabcol(gs2)
        mlab(trial) mlabpos(1) )
       (scatter risk ldl if arm==2 & (trial==17 | trial==18)
        , ms(i) mlabsize(vsmall) mlabcol(gs2)
        mlab(trial) mlabpos(12) )

xscale( range(1 , 5.4) )
xtitle(LDL-Cholesterol (mmol/L), size(medsmall) m(t+1) )
ytittle(Major Cardiac Event (%) , size(medsmall) m(r+1) )
xlab(1(1)5 , labsize(small) )
ylab(0"0%" 10"10%" 20"20%" 30"30%" , labsize(small) )

legend( order(1 2 3 4 5 6)
       lab(1 "Intervention/primary")
       lab(2 "Intervention/secondary")
       lab(3 "Intervention/both")
       lab(4 "Control/primary")
       lab(5 "Control/secondary")
       lab(6 "Control/both")
       size(vsmall)
       rowgap(*1.4)
       pos(11)
       col(1)
       ring(0) )

saving(lipid.gph, replace)
;
#delimit cr
```

```

*****
*           BROMLEY CORONARY REGISTER - KM SURVIVAL PLOT
*****
use bromreg , clear
#delimit ;
twoway(connected km_sf _t if diagexp==1 & _t<6 , ms(p) c(J) sort )
      (connected km_sf _t if diagexp==2 & _t<6 , ms(p) c(J) sort )
      (connected km_sf _t if diagexp==3 & _t<6 , ms(p) c(J) sort )
      (connected f fu if fu<6 , sort ms(p) c(J) clc(red) )

      ,
      xtitle("Time since diagnosis (yrs)"
            , m(t+2) size(medsmall) color(gsl) )
      ytitle("Cumulative proportion - all cause death"
            , m(r+1) size(medsmall) color(gsl) )

      ylab( 0(0.1)0.3 , format(%2.1f) labsize(small) )
      legend(off)
      xscale( range(0 7.3) )
      xlab(0 (1) 6 , labsize(small) )
      ylab( 0 (0.1) 0.3 , labsize(small) format(%2.1f) )
      xscale( r(0,7.2))
      xmtick(0(1)6)
      text(0.12 6 "New Exertional" "Angina"
            , color(dknavy) place(e) just(left) size(small))
      text(0.09 6 "Age-sex matched" "expected mortality"
            , color(red) place(e) just(left) size(small))
      text(0.22 6 "AMI"
            , place(e) size(small) color(dkgreen) just(left) )
      text(0.2 6 "Unstable Angina"
            , place(e) size(small) just(left) color("120 25 15"))

      text(-0.075 -0.79 "Number at risk" , size(medsmall) place(e) )
      text(-0.1 -0.75 "NEA" , size(small) place(e) )
      text(-0.12 -0.74 "UA" , size(small) place(e) )
      text(-0.14 -0.75 "AMI" , size(small) place(e) )

      text( -0.1 0.1 "284" , size(small) place(w) )
      text( -0.12 0.1 "79" , size(small) place(w) )
      text( -0.14 0.1 "174" , size(small) place(w) )
      text( -0.1 1.1 "276" , size(small) place(w) )
      text( -0.12 1.1 "75" , size(small) place(w) )
      text( -0.14 1.1 "154" , size(small) place(w) )
      text( -0.1 2.1 "271" , size(small) place(w) )
      text( -0.12 2.1 "75" , size(small) place(w) )
      text( -0.14 2.1 "148" , size(small) place(w) )
      text( -0.1 3.1 "266" , size(small) place(w) )
      text( -0.12 3.1 "73" , size(small) place(w) )
      text( -0.14 3.1 "146" , size(small) place(w) )
      text( -0.1 4.1 "263" , size(small) place(w) )
      text( -0.12 4.1 "71" , size(small) place(w) )
      text( -0.14 4.1 "143" , size(small) place(w) )
      text( -0.1 5.1 "256" , size(small) place(w) )
      text( -0.12 5.1 "68" , size(small) place(w) )
      text( -0.14 5.1 "140" , size(small) place(w) )
      text( -0.1 6.1 "155" , size(small) place(w) )
      text( -0.12 6.1 "35" , size(small) place(w) )
      text( -0.14 6.1 "90" , size(small) place(w) )

      graphreg( m(b+15) )
      saving(bromreg , replace)
;
#delimit cr

```

```

*****
*           AMBULATORY BLOOD PRESSURE: CONNECTED LINE PLOTS
*****
use amb_bp , clear

#delimit ;
tway (connected sbp dbp obs,  msize(small small) mc(red blue) clc(red blue) )
      (ksm sbp obs , clc(red) )
      (ksm dbp obs , clc(blue) )
      ,
      xlab( 1(4)23 , value)
      legend( order(1 2) )
      ytitle(Blood pressure (mmHg) , m(r+1) )
      xtitle(Time of day , m(t+1) )
      saving(abp1, replace)
;
#delimit cr
pause

#delimit ;
tway (connected sbp dbp obs,
      mcolor(blue red)
      clcolor(blue red)
      msize(small small)
      yaxis(1 2) )
      (function y=129 ,
        range(1 15.5) clcolor(blue) )
      (function y=88 ,
        range(1 15.5) clcolor(red) )
      (function y=115 ,
        range(15.5 24) clcolor(blue) )
      (function y=73 ,
        range(15.5 24) clcolor(red) )
      ,
      title("Ambulatory Blood Pressure hourly averages" ,
            size(medsmall) )
      xtitle(Time , margin(t+2) )
      ytitle("Blood Pressure (mmHg)" , m(r+2) axis(1) )
      xlab( 1(4)23 , valuelabel )
      ylab( , labsize(small)
            angle(hori)
            axis(1) )
      ylab( 73 "73 Mean DBP Night"
            88 "88 Mean DBP Day"
            115 "115 Mean SBP Night"
            129 "129 Mean SBP Day"
            , labsize(small)
            angle(hori)
            axis(2) )
      xline(15.5 , lcolor(gs12)
            lp(dash) )
      legend( off )
      saving(abp2, replace)
;
#delimit cr
pause

```

```

*****
*           META ANALYSIS OF LIPID LOWERING TRIALS - RANGE PLOTS
*****
use lipid2 , clear
#delimit ;
twoway (rspike low up trial , hori clc(gs8) )
      (scatter trial rr if trial>0 [fw= w] , ms(s) mcol(gs8) )
      (scatter trial rr if trial== -1 [fw=w] , ms(D) mcol(gs3) )
      ,
      ytitle(" ")
      xlab( , format(%2.1f) )
      legend(off)
      xscale( log)
      yscale( range(-1.25 16.25) )
      ylab(-1 1(1)17 , value labsize(small) angle(hori) nogrid )
      xline( 0.77 , lp(dash_dot) )
      xline( 1 , lp(dash) lc(gs7) )
      xtitle( "Risk Ratio" , size(medsmall) m(t+2) )
;
#delimit cr
pause

```

```

*****
*           FEV OVER TIME PLUS NUMBER IN STUDY - USING 2 Y AXIS
*****
use fev , clear

#delimit ;
graph twoway (rcap lfefv ufev month2 , blcolor(blue) yaxis(1) )
      (scatter mfev month2 if trt==1 , ms(S) mc(red) yaxis(1) )
      (scatter mfev month2 if trt==2 , ms(S) mc(green) yaxis(1) )
      (connected nfev month2 if trt==1 ,
        yaxis(2) c(L) clcol(blue) ms(s)
        msize(*0.7) mcol(red) mlab(nfev)
        mlabpos(12) mlabcol(gs3) mlabsize(*0.8) )
      (connected nfev month2 if trt==2 ,
        yaxis(2) c(L) clcol(blue) ms(s)
        msize(*0.7) mcol(green) mlab(nfev)
        mlabpos(6) mlabcol(gs3) mlabsize(*0.8) )
      ,
      xlabel(0"Baseline" 6(6)36 , labsize(small) )
      xscale( range(-1 38) )
      xtitle(Follow-up time (months) , m(t+2) size(small) )
      legend(off)
      yscale( range(1 1.5) axis(1) )
      ylabel(1.2(0.1)1.5 , axis(1) labsize(small) )
      yscale(range(0 1500) axis(2) )
      ylabel(0(100)400 , angle(hori) axis(2) labsize(small) grid )
      ytitle("Mean FEV (95% CI)" , m(r+2 b+25) axis(1) bexpand size(small) )
      ytitle(Number of subjects , m(l+2 b+2) axis(2) just(left) bexpand
size(small) )
      saving(fev , replace)
;
#delimit cr
pause

```

```

*****
*           WAIST CIRC AND BODY FAT % - TWOWAY AREA PLOTS
*****

use dataset1 ,clear
gen wcsq=wc_cm^2
regress bfp wc_cm wcsq
predict fval
predict stdp , stdp
gen low=fval-1.96*stdp
gen up=fval+1.96*stdp

pause
#delimit ;
graph twoway (rarea up low wc_cm
              , sort bcol(ltblue) )
              (scatter bfp wc_cm
              , ms(+) msize(tiny) )
              (line fval wc_cm , sort)
              ,
              ytitle(BFP (%))
              , size(medsmall) )
              xtitle(Waist circumference (cm))
              , size(medsmall) )
              xscale( alt range(50 140) )
              yscale( alt range(15 60) )
              xlab( , labsize(small) grid gmax )
              ylab( 20(10)60 , labsize(small) )
              legend(off)
              saving(wc_bfp, replace)
;
#delimit cr
pause

```

```

*****
*           TWOWAY HISTOGRAMS
*****
graph twoway histogram wc_cm
pause

#delimit ;
graph twoway histogram bfp , fraction
        xscale( alt reverse)
        yscale( range(15 60) )
        ylab(20(10)60 , labsize(small) )
        xlab( , labsize(small) )
        horizontal
        saving(bfp , replace)

;
#delimit cr

#delimit ;
graph twoway histogram wc_cm , fraction
        yscale(alt reverse)
        xscale( range(50 140) )
        ylab( , nogrid labsize(small) )
        xlab( 60(20)140
              , grid gmax labsize(small) )
        saving(wc , replace)

;

#delimit ;
graph twoway histogram bfp , fraction
        xscale( alt )
        yscale( range(15 60) )
        ylab(20(10)60 , labsize(small) )
        xlab( , labsize(small) )
        horizontal

;

#delimit ;
graph twoway histogram smoke ,
        discrete xlab(1 "Never" 2 "Ex" 3"Current" ) gap(5) freq
        title(Distribution of Smoking at baseline)
        ytitle(Frequency , margin(r+2) )
        xtitle(Smoking status , margin(t+2) )

;
#delimit cr
pause

```

```

*****
*           Association between BMI and CRP - BOX PLOTS
*****

use dataset1, clear

graph hbox crp , over(bmi4)
pause

#delimit ;
graph hbox log_crp ,
    over(bmi4 ,
        label( labsize(small)
              labcolor(green) ) )
    over(agegroup ,
        relabel(1"<40 yrs"
                2"40-49 yrs"
                3"50+ yrs" ) )
        marker(1, msymbol(smplus) )
    title("Distribution of log CRP by BMI and age" ,
        color(black)
        size(medium) )
    ytitle( Log(CRP(mM)) ,
        size(medsmall) )
    yscale( log)
    note(Source: UKWHS - Phase 2)
    saving(box, replace)
;
#delimit cr
pause

*****
*           Using/exporting/combining previously created/saved graphs
*****

graph use wc_bfp
pause

graph export wc_bfp.emf , replace
pause

graph use bfp
graph use wc
graph use wc_bfp

#delimit ;
graph combine bfp.gph wc_bfp.gph wc.gph ,
    hole(3)
    imargin(0 0 0 0)
    saving(combl, replace)
;
#delimit cr
pause

```

```

*****
*                POPULATION PYRAMID - TWOWAY BAR GRAPHS
*****
use popn.dta , clear
#delimit ;
graph twoway
    (bar maletotal agegrp , horizontal bcolor(blue*0.5) )
    (bar femtotal agegrp , horizontal bcolor(red*0.5) )
    (scatter agegrp z , mlab(agegrp) ms(i)
        mlabpos(0) mlabcolor(black) )
    ,
    xlab(-12"12"-10"10"-8"8"-6"6"-4"4" -2" "0" "2" " 4(2)12 ,
        noticks grid gmin gmax
        glcolor(gs13)gpattern(dash) )
    yscale( off )
    xscale( noline
        titlegap(-3) )
    plotregion( style(none)
        margin(zero) )
    legend( order(1 2)
        label(1 Males)
        label(2 Females)
        symxsize(*0.25) )
    title("US Population by Sex and Age: 2000",
        size(medium) )
    xtitle(Population in millions )
    saving(popn , replace)
;
#delimit cr
pause

*****
*                NORMAL DENSITY FUNCTION: GRAPH TWOWAY FUNCTION
*****

#delimit ;
graph twoway (function y=normden(x)
    , range(-4 -1.96) bcolor(gs12) recast(area) )
    (function y=normden(x)
    , range(1.96 4) bcolor(gs12) recast(area) )
    (function y=normden(x)
    , range(-1.96 1.96) bcolor(gs14) recast(area) )
    (function y=normden(x)
    , range(-4 4)
        clstyle(foreground)
        dropline(-1.96 0 1.96) )
    ,
    plotregion( style(none) icolor(white) )
    yscale(off)
    xscale(noline)
    ylab( , nogrid )
    legend(off)
    xlabel(-4 "-4 sd" -3 "-3 sd" -2 "-2 sd"
        -1 "-1sd" 0 "mean" 1 "1 sd"
        2 "2 sd" 3 "3 sd" 4 "4 sd" )
    xtitle("")
    saving(function , replace)
;
#delimit cr
pause

```

```
*****
*
*                               PIE CHARTS
*
*****
```

```
use bmi_long_pl_p2.dta , clear
#delimit ;
graph pie , over(bmi)
    by(phase ,
        title("Distribution of BMI at Phase 1 & Phase 2 of UKWHS"
            , size(medsmall) )
        note("Phase 1: 1988-91" "Phase 2: 1993-97" )
    )
    plabel(_all percent , format(%2.0f) )
    pie(1 , color(green*0.75) )
    pie(2 , color(green*0.5) )
    pie(3 , color(orange*0.75) )
    pie(4 , explode color(red*0.75) )
    legend( title("BMI (kg/m-sq)" ,
        size(medsmall) )
        rows(1) )

;
#delimit cr
pause
```